

Vakiokertoimisen lineaariyhtälön karakteristisen yhtälön kompleksiset juuret

Tarkastelun kohteena olkoon kertalukua n oleva vakiokertoiminen homogeeninen differentiaaliyhtälö:

```
> n:= 3;
```

```
n:= 3
```

```
> diffyht:= sum('a| |k'*diff(y(x),[x$k]),k=0..n)=0;
```

$$\text{diffyht} := a_0 y(x) + a_1 \left(\frac{d}{dx} y(x) \right) + a_2 \left(\frac{d^2}{dx^2} y(x) \right) + a_3 \left(\frac{d^3}{dx^3} y(x) \right) = 0$$

Karakteristinen yhtälö saadaan sijoittamalla yhtälöön yrite $y = e^{rx}$ ja jakamalla sijoittamisen jälkeen eksponenttitekijä pois. Samaan tulokseen päästään korvaamalla differentiaaliyhtälössä derivaatat vastaavilla muuttujan r potensseilla:

```
> {seq(diff(y(x),[x$k])=r^k, k=0..n)}:  
karaktyht:= subs(%,diffyht);
```

$$\text{karaktyht} := a_0 + a_1 r + a_2 r^2 + a_3 r^3 = 0$$

Kyseessä on polynomiyhtälö, jonka juuret r_1, r_2 jne. antavat differentiaaliyhtälön perusratkaisut

$y = e^{(r_1 x)}$, $y = e^{(r_2 x)}$ jne. Oletetaan, että yhtälöllä on kompleksinen juuripari $r_1 = \alpha + \beta I$, $r_2 = \alpha - \beta I$, ja pyritään osoittamaan, että tällöin vastaaviksi perusratkaisuiksi kelpaavat $e^{(\alpha x)} \sin(\beta x)$ ja $e^{(\alpha x)} \cos(\beta x)$. Koska $r_1 = \alpha + \beta I$ ja $r_2 = \alpha - \beta I$ ovat juuria, seuraavat ehdot toteutuvat:

```
> komplehdot:= {subs(r=alpha+I*beta,karaktyht),  
subs(r=alpha-I*beta,karaktyht)}:  
simplify(%);
```

```
{
```

$$a_0 + a_1 \alpha + a_1 \beta I + a_2 \alpha^2 + 2 I a_2 \alpha \beta - a_2 \beta^2 + a_3 \alpha^3 + 3 I a_3 \alpha^2 \beta - 3 a_3 \alpha \beta^2 - a_3 \beta^3 I = 0,$$

$$a_0 + a_1 \alpha - a_1 \beta I + a_2 \alpha^2 - 2 I a_2 \alpha \beta - a_2 \beta^2 + a_3 \alpha^3 - 3 I a_3 \alpha^2 \beta - 3 a_3 \alpha \beta^2 + a_3 \beta^3 I = 0$$

```
}
```

Yhtälöissä esiintyvät vakiot ovat kaikki reaalisia, jolloin periaatteessa kahdesta kompleksisesta ehdosta saadaan neljä reaalista ehtoa asettamalla kummankin reaali- ja imaginaariosa erikseen =0. Erilaisia ehtoja saadaan kuitenkin vain kaksi, kuten pitääkin: Jos nimittäin reaalikertoimisella polynomiyhtälöllä on juurena $\alpha + \beta I$, myös liittoluku on juuri.

Imaginaariluvun reaaliosa ja imaginaariosa saadaan poimittua komennoilla **Re** ja **Im**. Koska

Maple ei tiedä ovatko muuttujat reaalisia vai kompleksisia, on sievennyskomennon **simplify** yhteydessä käytettävä parametria **assume=real**, joka määrittää muuttujat reaalisiksi.

```
> {Re(lhs((komplehdot[1])))=0, Im(lhs((komplehdot[1])))=0}:
ehdot:= simplify(%, assume=real);
```

```
ehdot:=
```

$$\{a_1 \beta + 2 a_2 \alpha \beta + 3 a_3 \alpha^2 \beta - a_3 \beta^3 = 0, a_0 + a_1 \alpha + a_2 \alpha^2 - a_2 \beta^2 + a_3 \alpha^3 - 3 a_3 \alpha \beta^2 = 0\}$$

Sijoitetaan tutkittavat perusratkaisut differentiaaliyhtälöön:

```
> sij:=
{algsubs(y(x)=exp(alpha*x)*sin(beta*x), diffyht),
algsubs(y(x)=exp(alpha*x)*cos(beta*x), diffyht)}:simplify(%);
{e(α x) (a2 α2 sin(β x) + 2 a2 α cos(β x) β - a2 sin(β x) β2 + a0 sin(β x) + a1 α sin(β x)
+ a1 cos(β x) β + a3 α3 sin(β x) + 3 a3 α2 cos(β x) β - 3 a3 α sin(β x) β2 - a3 cos(β x) β3)
= 0, e(α x) (a2 α2 cos(β x) - 2 a2 α sin(β x) β - a2 cos(β x) β2 + a0 cos(β x)
+ a1 α cos(β x) - a1 sin(β x) β + a3 α3 cos(β x) - 3 a3 α2 sin(β x) β - 3 a3 α cos(β x) β2
+ a3 sin(β x) β3) = 0 }
```

Ongelmana on, toteutuvatko nämä yhtälöt, kun oletetaan, että α ja β toteuttavat edellä johdetut ehdot:

```
> simplify(sij,ehdot);

{0 = 0}
```

Yhälöt näyttävät olevan tosia, ja vastaavat perusratkaisut siis todella ovat $e^{(\alpha x)} \sin(\beta x)$ ja $e^{(\alpha x)} \cos(\beta x)$.

Symbolisissa laskentajärjestelmissä kuten Maplessa on suhteellisen hyvät työkalut (usean muuttujan) polynomiehtojen käsittelyyn; taustalla ovat ns. *Gröbnerin kannat*. Probleeman monimutkaistuesssa laskennan raskaus kuitenkin kasvaa nopeasti.

Kompleksisten juurten tapausta voidaan käsitellä myös toisin. Koska kompleksiluvut toteuttavat samanlaiset laskusäännöt (derivointisäännöt mukaanluettuina) kuin reaaliluvutkin, voidaan ajatella, että kirjoitetaan differentiaaliyhtälön yleinen ratkaisu kompleksisten eksponenttifunktioiden avulla:

```
> komplratk:= _C1*exp((alpha+I*beta)*x)+_C2*exp((alpha-I*beta)*x);
komplratk:= _C1 e((α+β I) x) + _C2 e((α-β I) x)
```

Tämä kehitetään käyttäen eksponenttifunktion laskusääntöä $e^{(x+y)} = e^x e^y$, joka on voimassa myös kompleksitapauksessa, ja sen jälkeen sovelletaan Eulerin kaavaa $e^{(I u)} = \cos(u) + I \sin(u)$. Laskun

voi tehdä käsin tai käyttämällä sopivasti *Maplen* komentoja.

Convert-komento muuntaa lausekkeen kaikki eksponenttimerkinnät trigonometrisiksi. Koska emme kuitenkaan halua muuntaa termejä, jotka sisältävät $e^{(\alpha x)}$ trigonometrisiksi, on tehtävä muunnos takaisin eksponenttimuotoon.

```
> convert(komplratk, trig):  
trigratk:= subs(convert(exp(x*alpha), trig)=exp(x*alpha), %):  
expand(%);
```

$$\begin{aligned} & _C1 e^{(\alpha x)} \cos(\beta x) + _C1 e^{(\alpha x)} \sin(\beta x) I + _C2 \cosh(\alpha x) \cos(\beta x) \\ & - _C2 \cosh(\alpha x) \sin(\beta x) I + _C2 \sinh(\alpha x) \cos(\beta x) - _C2 \sinh(\alpha x) \sin(\beta x) I \end{aligned}$$

Tämä todellakin näyttää sisältävän haluttua muotoa $e^{(\alpha x)} \sin(\beta x)$ ja $e^{(\alpha x)} \cos(\beta x)$ olevia termejä. Näiden kertoimet ovat

```
> coeff(trigratk, sin(beta*x)):
sinkerroin:=coeff(%, exp(alpha*x));
sinkerroin := \_C1 I
> coeff(trigratk, cos(beta*x)):
coskerroin:=coeff(%, exp(alpha*x));
coskerroin := \_C1
```

Ratkaisu voidaan sieventää antamalla näille kertoimille nimet $_C3$ ja $_C4$:

```
> simplify(trigratk, {sinkerroin=\_C3, coskerroin=\_C4});
\_C2 cosh(\alpha x) cos(\beta x) - \_C2 cosh(\alpha x) sin(\beta x) I + \_C2 sinh(\alpha x) cos(\beta x)
- \_C2 sinh(\alpha x) sin(\beta x) I + (e^{(\alpha x)} cos(\beta x) + e^{(\alpha x)} sin(\beta x) I) \_C4
>
```

$_C1$ ja $_C2$ sekä toisaalta $_C3$ ja $_C4$ ovat symmetrisessä asemassa. Jos kumpi tahansa pari valitaan mielivaltaisesti, toinen pari määräytyy.

Linkit

[vakiokertoiminen homogeeniyhtälö](#)